```
model{

for(p in 1:N){
for(i in 1:nit){

logit(p.slow[p,i]) <- a.slow[i]*theta.slow[p] - b.slow[i]                 # Equation 5
logit(p.fast[p,i]) <- a.fast[i]*theta.fast[p] - b.fast[i]                 # Equation 6
mu.rt[p,i]<- icept[i] + td[i]* tau[p]                                     # Conditional mean of Eq 2
lrt[p,i]~dnorm(mu.rt[p,i],prec[i])                                        # Distribution of lnT
logit(pi[p,i])<- time[i]*(lrt[p,i]- td[i]*tau[p] - icept[i]-marg[i])      # Equation 9 with epsilon given
                                                                          # by Equation 11
prob[p,i] <- pi[p,i]*p.slow[p,i]+(1-pi[p,i])*p.fast[p,i]                  # Equation 4
x[p,i]~dbern(prob[p,i])                                                   # Distribution of X

}}

mu_ze[1]<-0
mu_ze[2]<-0
mu_ze[3]<-0

# Below the variances of the latent variables are constrained to be equal to 1, see Equation 10

for(i in 1:3){
for(j in 1:3){
	cor_mat[i,j]<-gamma[i] * gamma[j] + step(-1*abs(i-j)) * (1-gamma[i] * gamma[j])
}
	gamma[i]~dunif(-1,1)                                              # prior on gamma
}
	prec_mat[1:3,1:3]<-inverse(cor_mat[1:3,1:3])                      # precision matrix

for(p in 1:N){
	lats[p,1:3]~dmnorm(mu_ze[],prec_mat[,])                           # normal distribution for latent
	                                                                  # variables
	tau[p]<-lats[p,1]
	theta.slow[p]<-lats[p,2]
	theta.fast[p]<-lats[p,3]
}

# below, the prior distributions are specified as discussed in the section “estimation” on page 6

for(i in 1:nit){
	icept[i]~dnorm(0,.1)
	prec[i]<-1/exp(sigma[i])
	sigma[i]~dnorm(0,.1)

	b.slow[i]~dnorm(0,.1)
	b.fast[i]~dnorm(0,.1)
	time[i]~dnorm(0,.1)I(0,)
```

```
        marg[i]~dnorm(0,.1)

}


for(i in 1:nit){
td[i]~dnorm(0,.1)
a.slow[i]~dnorm(0,.1)I(.01,)
a.fast[i]~dnorm(0,.1)I(.01,)

}


}
```